
Matériels nécessaires  
 

1) Un AMIGA ou l'émulateur WINUAE 

2) Une Carte ACTION REPLAY (ou ça ROM Image) selon configuration utilisé. 
3) Le jeu Original Another World ou son image CAPS française (SPS 2377) 

4) Le logiciel Xcopy-pro en disquette ou son image disk. 
5) L’assembleur ASM-One 

6) Les connaissances pour utiliser un l’assembleur ci-dessus ! 

 

 

Général Info 
Ce tutoriel Français est basé sur le tutoriel original mis en ligne sur le site flashtro.com. 

Ce document n'est pas une traduction mot par mot de celui-ci mais plus une nouvelle version. 
Suivit pas à pas avec des nouvelles informations et parfois une nouvelle approche. 

 
 

 

 
 

 
 

Bon Tuto.              
        Gi@nts 

Agencement des disquettes Amiga v1.1 
 

En France : 
 
On utilise des termes comme : piste, bloc, secteur, face… 

Piste : 0 à 79 Certaine disquette pousse jusqu’à 81 voire 82 pistes mais le standard reste quand même 80 pistes (de 0 à 79) 

Face : 0/1 ou 1/2 ou A/B, Dessus ou dessous tout simplement. Sur Amiga nous avons deux faces utilisées sur 99% des jeux. 
 

Chaque piste, pour un format standard ‘AmigaDOS’ est composée de plusieurs bloc ou secteur, en général 11 par face. 
Le terme piste peut désigner l’ensemble d’une piste (les deux ‘side’ du disque), ou uniquement une ‘side’ d’une piste. 

Une piste standard amigados est découpée en plusieurs partie appelé bloc, secteur, sector.  
 

 
 

 
 

 

 

Dans d’autre pays :  

 

On utilisera d’autre terme, comme sector, keys, tracks, cylindre, head… 
 

Le terme track par exemple que l’on aurait vite fait de traduire ‘piste’ ne colle pas forcément à notre description française. 
En général, le terme tracks désigne toujours une position sur la disquette mais elle va de 0 à 159 (soit 160 tracks) 

Le maximum étant 160 et non 80 car on a deux faces bien sûres, en fait, elle correspond à une piste sur une face. 
 

Il peut néanmoins arriver que l’on utilise dans des tuto anglo-saxon le terme tracks dans le sens ‘piste’ en français 
(donc de 0 à 79 et non de 0 à 159). Mais en règle générale, il a plutôt une plage de 0 à 160. 

 

C’est le terme cylindre qui ‘colle’ plus à notre définition française de piste. 
En effet, il est courant d’utiliser le terme cylindre pour désigner une position sur la disquette de 0 à 79. 

 



Le terme sector ou key quant à lui corresponds au terme français bloc ou secteur. 
 

Sur une disquette au format Amigados, nous avons 880ko et nous avons 11 secteurs par face, par piste. 
 

La taille d’une piste ayant une valeur physiquement maximum. 
Le nombre maximum de sector sur une piste dépends assez logiquement de la taille de ses sectors. 

 
Pref…beaucoup de terme qui ne sont pas forcément utilisé dans leur sens propre, le mieux est de lire un tuto 

et de comprendre quel sens l’auteur a voulu leurs donner. 
 

Il existe aussi un autre type d’appellation utilisé par exemple par le logiciel MFM-Warp  de Ferox* 

*C’est un programme qui scan le disque en bas niveau et essaye d’en réaliser une copie. 
 

Track Calcul Résultat Format  utilisé sous MFMWarp 

0 0/2 0 et pair 0.0 

1 1/2 0 et impair 0.1 

2 2/2 1 et pair 1.0 

3 3/2 1 et impair 1.1 

156 156/2 78 et pair 78.0 

157 157/2 78 et impair 78.1 

158 158/2 79 et pair 79.0 

159 159/2 79 et impair 79.1 
 

 

On notera que : 

 
 

Le premier secteur (secteur 0) appelé aussi bootbloc commence sur la lowerSide en piste 00 

et se fini en piste 79 sur le upperside 
 

En tracks c’est le même système sauf que l’on terminera en Track 179 et non 79. 
 

La piste Zero est celle situé le plus à l’extérieure du disque. 

 
Le 1er secteur logique, donc le premier bloc sur la disquette, se trouve piste 0 secteur 0 

Les bloc se suivent physiquement mais ne sont pas forcément ordonnée, on parle aussi d’entrelacement. 
 

Le bloc 11 (si on part de 0 bien sur) n’est pas le 1er secteur de la seconde piste mais le 1er secteur de la face suivante. 
(voir image ci-dessus) 

 
En format Amigados, la taille d’un secteur est de 512 octets 

Ce qui nous donnes comme taille disponible : 512*11 secteurs*80 pistes*2 faces = 901 120 octets soit 880Ko 
Une ‘track’ AmigaDos a une taille de 512 * 11 = 5632 en décimal soit $1600 octets 

 
 

Mise en application sous l’AR : 

 
Il existe deux commandes sous l’AR qui permettent de charger sauver des pistes, à savoir : RT et WT 
Elles fonctionnent pareil. 

L’une permet la lecture, l’autre l’écriture. 
 

#RT alias Read Track. Permet le chargement de donnée située sur la disquette vers la mémoire. 
#la première valeur sera la track de départ [0 à 159] à indiquer en hexa.                   /!\ ne pas confondre avec piste 

 
#La seconde valeur sera le nbr de demi track à copié à partir de là. 

 
#WT alias Write Track. Permet la sauvegarde de donnée située en mémoire vers la disquette. 

 
 

Exemples :  
 
RT 20 1 50000 
Start Track = $20 et taille à lire = 1 

On copiera donc la piste !16 (en décimal) side 0 en mémoire $50000 

 
Oui car 20 est donné en hexa, ce qui nous donnes !32 en décimal mais il indique une track (de 0 à 159) PAS en piste. 

Pour avoir l’équivalent en piste on divisera donc par 2 (car deux faces). 
 

$20/2=$10 = !16 (en décimal donc) et comme il n’y a pas de retenu on est sur la face0. 
 

 
RT 21 2 

Start Track = $21 et taille à lire = 2 
On copiera la piste !16 side 1 et la piste !17 side 0 en mémoire 50000 

 
21 est donné en hexa donc $21 = !33 en décimal. 

33/2 = 16.5 ,donc piste 16 side 1 et comme on continue à lire/copier les donnes (taille à lire =2), on continue la copie. 
On change donc de track car on est déjà sur la face 1 (il existe que 2 faces sur une disquette) 

On arrive donc sur la prochaine track à savoir, piste 17 en side 0 puisque que c’est la première face au niveau structure la side 0. 

  



WinUAE 

 
Pour ceux qui utilisent winUAE pour ces tutoriels (j’imagine, la plupart des personnes), Je vous conseille fortement d’activer le son des lecteurs 

de disquette histoire d’entendre ce que le lecteur effectue comme accès. 

HOST -> SOUND -> FLOPPY DRIVE SOUND EMULATION - > DF0 Built-In 

 

 
 
 
Voir même, pour plus d’information. Par exemple afficher sur qu’elle face l’on se trouve, d’activer :  

Host -> Miscellaneous -> Native on-screen display AND RTG on-screen display 

 

 



Part 1 X-copy 

Time to check the protection 
On va dans un premier temps essayer de copier les disquettes originales à l’aide de Xcopy Pro. 
 

 

 
Bon…Que ce soit la disquette 1 ou 2, aucun problème pour la copier, on s’oriente donc vers une protection unique du genre ‘papier’. 

 
 

Rappel des codes d'erreur de Xcopy :  

 
1. Less or more than 11 sectors 

2. No sync found 
3. No sync after gap found 

4. Header checksum error 
5. Error in header/format long 

6. Data block checksum error 

7. Long track 
8. Verify error 

 
  



Part 2 Analyse de l’image IPF 

 

FILENAME 2377_Another_World_D2_AMIGA.ipf 

TYPE Floppy_Disk 

ENCODER CAPS(V1) 

FILE 2377(V1) 

DISK 2 

TRACK 00-83 

SIDE 0-1 

PLATFORM Amiga 

REVOLUTION 3 
 

 
Comme prévue, rien de spécial sur les deux disquettes. Un format Standard AmigaDOS sur toutes les pistes. 

 

 
 

 
 

 

 



Part 3 Protection papie : roue codeuse 

Le jeu utilise un système de protection papier. 
Une impression circulaire qui tourne sur une autre laissant apparaitre des codes aux endroits troués. 

 
 

 
  



Part 4 Let’s Rock’n Roll  

 
Nous avons détecté aucune protection sur les disquettes et n’avons rencontré aucune erreur lors de leurs copies, aussi, 

nous allons travailler dans ce tuto avec les disquettes de backup 😊 

 
Insérer votre backup de la disquette 1 d’Another World dans le lecteur et Allumer votre Amiga. 

Après un cours chargement, un système de protection papier nous est demandé (en l’occurrence un disque carton de code) 

 

 
 

On va chercher en mémoire les adresse de code qui font appel à $BFE001 

$BFE001 qui rappelons-le, est l’adresse du CIA-A qui gère le registre pra (/FIR1 /FIR0 /RDY /TK0 /WPRO /CHNG /LED OVL) 

Dans notre cas, on s’intéresse à l’appuie sur le bouton 1 (FIR1) donc, bits 7 du registre en question. 

Petit piqure de rappel : 

 

 
 

Entrer dans votre AR et taper :  

#FA, alias Find for Addressing Upcode, On va chercher une instruction qui fait appel à une adresse mémoire. 
fa BFE001 

 

 
 
On va s’intéresser en 1er temps à la 1er adresse trouvée, à savoir : $2145C 
Taper la commande suivante pour désassembler le code à partir de l’adresse indiqué. 
#D, alias Désassemble 
d 2145C 

 

 
  



02145C TST.B    00BFE001  #Test de l’adr $BFE001, Flags mis à jour en conséquence. 

021462 BMI      00021470  #Si résultat négatif (flag N=1), alors on branche en $21470. 

021464 MOVE.W   #1,1F4(A1) #Sinon, on continue le code. 

02146A ORI.W    #80,1FS(A1) 

021470 TST.W    00022FA8  #Test de l’adr $22FA8, Flags mis à jour en conséquence. 

021476 BEQ      00021480  #Si résultat est égal (flag Z=1), alors on branche en $21480. 

 

Afin d’être sûr que l’on se trouve bien dans la partie du code qui nous intéresse, nous allons insérer un Breakpoint. 

 
#BS, alias BreakPoint. Permet, dès que l'adresse mémoire indiquée est atteinte, d'effectuer un arrêt du code. 
Taper : BS 21464 

Breakpoint inserted Ready. 

Pour info, c’est à +8 de la 1er adresse trouvée par la commande fa (02145C+8=21464) 

 

On retourne à l’exécution du jeu.  
#X permet le retour au code Amiga en court. 
Taper : X 

Rien de spécial se passe, mais lorsque l’on appuie sur le bouton FIRE, Bingo 

Notre BreakPoint est atteint et on entre automatiquement dans l’AR  
 

Taper : D puis R 

 

 
 

On a bien atteint comme prévu notre BreakPoint et on est bien à l’adresse mémoire $21234. 

En regardant la valeur de A1, on peut voir qu’elle indique $7C648 
Si on reprend le code, on peut voir que :  

 

021464 MOVE.W   #1,1F4(A1) #Ajoute 1 (en Word) à (l’adresse contenue de A1 + 1F4)  

     #Donc 7C650+1F4=7C844 

02146A ORI.W    #80,1FS(A1) #Humm, on travaille encore avec A1… 

 

On va donc s’intéresser à elle pour commencer. 
#M alias memory read. Permet de voir les données en mémoire au format HEXA/ASCII. 
Taper : M 7C650 

 

 
 
On va laisser le code continuer normalement et donc, supprimer le BreakPoint que l’on avait mis. 

#BD, alias BreakPoint Delete. Permet de supprimer un breakpoint préalablement posé. 
Taper : BD 21464 puis X 

  



On est donc revenu sur le menu de la protection du jeu. 
Le code se poursuit et vue que l’on avait appuyée sur Fire, un symbole est validé. 

Sans appuyer sur quoi que ce soit, Entrer à nouveau dans l’AR et allons jeter un coup d’œil de nouveau à cette adresse. 
 
Taper : M 7C650 

 
On notera quelques changements (ici surligné en jaune) 

 
 
Taper : X 

et appuyer sur Fire pour sélectionner un code puis, Entrer à nouveau dans l’AR 

Allons jeter un coup d’œil encore une fois à notre adresse de prédilection 😊 

Taper : M 7C650 

 

On notera quelques changements par rapport au deux sélections précédente (ici surligné en violet) 

 
 

Ce qui nous donnes :  
 

Adresse Mémoire 1er sélection 2eme sélections 3em sélections 

7C655 00 01 02 

7C68F 00 0C 0C 

7C695 00 01 02 

7C6B5 00 61 62 

7C6BE -> 7C6BF 45  21 C2  93 7E  7E 

7C6EE -> 7C6EF 02  F3 04  83 06  DB 
 

 
$7C655 et $7C895 pourraient être un compteur du nombre de symboles sélectionnés… 

On va suivre ces deux adresses et vérifier ceci.  
 

  



Part 5 Adresse mémoire, Nombre de symbole sélectionné  

 
Retournons à notre menu de protection du jeu, Taper : X 

Puis effacer les symboles sélectionnés à l’aide de l’icône DEL :  
 

 
 

Vous devez maintenant ne plus avoir de symbole sélectionné, entrer dans l’AR 
Taper : M 7C655 et M 7C695 

 

 
 
Taper : X 

Puis sélectionner un symbole et entrer dans l’AR. 
Taper : M 7C655 et M 7C695 

 

 
 
Taper : X 

Puis sélectionner un second symbole et entrer dans l’AR. 
Taper : M 7C655 et M 7C695 

 

 
 
Taper : X 

Puis sélectionner un troisième symbole et entrer dans l’AR. 
Taper : M 7C655 et M 7C695 

 

 
 
Taper : X 

Puis sélectionner enfin un quatrième symbole et entrer dans l’AR. 
Taper : M 7C655 et M 7C695 

 

 
 

 
Bon… Il est plus que probable qu’on a affaire ici à un compteur de symbole sélectionné. 

Adresse mémoire : $7C655 et $7C895 
  



 

Part 6 Analyse mémoire du 1er menu de protection  
 

Retournons au menu de protection du jeu. 
Taper : X 

 

Effacer les symboles sélectionnés à l’aide de l’icône DEL :  
 

 
 
A l’aide de notre rouge codeuse, entrer maintenant les bons symboles. 

Dans notre cas en N6 

 

 
 

Pour plus de facilité, on va remettre notre BreakPoint puis retourner au jeu. 
Entrer dans l’AR de nouveau et taper : BS 21464 puis X 

 

De retour au jeu, le curseur étant toujours sur l’icône OK, Appuyer sur FIRE 
Notre BreakPoint est logiquement atteint et on entre automatiquement dans l’AR.  

 
Allons jeter un coup d’œil en mémoire pour voir si changement il y a eu. 
Taper : M 7C650 

 

Ici, mise en évidence via des couleur les changements notables par rapport aux précédents ‘dumps’ 

 
 

Comme prévu on retrouve notre compteur de symboles à 03 (A1+$05, A1+$45) 

 
 

La valeur $012C est positionnée 3 fois (A1+$0E, A1+$4A, A1+$4E) 

 
 

 



La valeur $07 est elle aussi positionnée 3 fois (A1+$03, A1+$0B, A1+$47) 

 
 
Et si on regarde vraiment bien, on peut voir que 3 nouvelles valeurs correspondent avec 3 valeurs déjà présentes. 

Word $0016 en A1+$3C = Word $0016 en A1+$52 
Word $0006 en A1+$3E = Word $0006 en A1+$56 

Word $0009 en A1+$40 = Word $0096 en A1+$54 
 

Ici dans l’ordre 1-3-2 
 

 
 

Le $62 est passé à $69 (A1+$65) 
Un changement de valeur est aussi visible en A1+$6E et A1+$9E 

 
 

 
  



 

Part 7 Tableau de position des symboles 
 

Retournons au menu de protection du jeu. 
Taper : X 

Le code préalablement entré est validé et un second menu, identique, apparait pour demander encore une suite de symbole. 

On va s’intéresser pour l’instant à essayer de comprendre pourquoi le 03 c’est changé en 07 (A1+$03, A1+$0B, A1+$47) 
Déplacer Le curseur sur l’icône OK. 

 
Entrer dans l’AR et Taper :  

M 7C650+3 [Entrée]  [Esc] 

M 7C650+b [Entrée]  [Esc] 

M 7C650+47 [Entrée]  [Esc] 

 

 
 
Retournons au menu de protection du jeu. 
Taper : X 

Et déplacer maintenant le curseur sur l’icône DEL : 
 

 

 

Entrer de nouveau dans l’AR et Taper :  

M 7C650+3 [Entrée]  [Esc] 

M 7C650+b [Entrée]  [Esc] 

M 7C650+47 [Entrée]  [Esc] 

 

 

 

Humm, cela ressemble à la position du curseur. 
Vous pouvez recommencer l’opération en déplaçant le curseur sur chaque icone, vous devrez arrivez à la conclusion ci-dessous : 

 

 



Part 8 Analyse mémoire du 2em menu de protection  
 

Allons voir maintenant à quoi ressemble nos données en mémoire $7C650 
Taper : M 7C650 

 

 
 

 
Comme on pouvait si attendre, les données en A1+$3C, A1+$3E et A1+$40 ont changé, maintenant elles sont à 00 00 00 00 00 00 

Logiquement, les données en A1+$52, A1+$56, A1+$54 ont aussi changé, maintenant elles sont à 00 0A 00 15 00 13 
 

A l’aide de notre rouge codeuse, entrer maintenant les bons symboles. 
Dans notre cas en I11 

 

 
 
Notre BreakPoint en $21464 étant toujours d’actualité, déplacer le curseur sur l’icône OK et appuyer sur FIRE. 

Comme prévu, on entre automatiquement dans l’AR. 
 

Allons jeter un coup d’œil en mémoire pour voir les changements. 
Taper : M 7C650 

 
 

On regarde ça de plus prêt. 

 
 
On retrouve bien notre ‘logique’ précédente mais l’ordre a changé… 

Ici en ordre : 3-2-1 alors que dans la PART6 de ce tuto, on était en ordre : 1-3-2 
Nous y reviendrons dans le prochain chapitre. 

 
Si on retourne au code du jeu, le code se poursuit. 

Quelques petits chargements font leurs apparitions, deux logos s’enchaine suivie d’information sur le jeu le tout suivi 
d’un plus long chargement et l’animation d’intro se lance. 

  



Part 9 Création disquette de Test  
 

Après plusieurs reboot et essaie, on peut s’apercevoir qu’une fois les bons symboles entrés, l’ordre des DATA en A1+$3C, A1+$3E et A1+$40 
n’est pas toujours le même. Dés fois c’est dans l’ordre et des fois nom MAIS, toujours ordonnée en WORD. 

 

Cette histoire de changement d’ordre est ennuyeuse.     -_-‘ 
On est peut-être passé à côté de quelque chose comme une valeur qui donne l’ordre des DATA ?! 

 
Effectuer un reboot de votre Amiga (Contr+Alt gauche-Amiga-Amiga) 

Laisse le jeu se charger pour arriver au menu de sélection des symboles de code. 
Attendez quelques secondes pour que l’état mémoire se stabilise et entrer dans l’AR. 

Insérer une nouvelle disquette vierge dans le lecteur, nous allons sauver tout ça ! 
 

Taper : 

#format, Permet de formater une disquette au format AmigaDOS 
FORMAT save1 

Ready to format disk in drive DF0: (y/n)? y 

Disk ok 
 

#SA, alias SaveAllMemory. Permet donc de sauver toute la zone mémoire du programme en court. 
SA FULL 

 

#INSTALL, alias BOOTBLOCK INSTALL, permet d'installer un secteur de boot sur l'unité indiquée, 0=DF0, 1=DF1 
INSTALL 

Ready to install disk in drive DF0: (y/n)? y 

Disk ok 
 

#SLOADER, alias SaveLoader. Permet de sauver le code de chargement interne de sauvegarde de la MKIII vers un exécutable. 
SLOADER  

 

#MAKEKDIR, alias MakeDirectory. Permet de créer un répertoire. 
MAKEDIR s 

 

#CD, alias ChangeDirectory. Permet de changer le répertoire en court, d’entrer dans un répertoire  
CD s 

 

#N, alias memory read. Permet de voir/écrire les données en mémoire au format uniquement ASCII. 
N 100 

.000100 Aload FULL   [RETURN4] 

.000140 [ESC] 

 

 
 

#SM, alias SaveMemory. Permet donc de sauver une zone mémoire vers un fichier. 
SM startup-sequence, 100 100+10 

 

 

Effectuer un reboot de votre Amiga avec toujours notre nouvelle disquette fraichement crée dans le lecteur. 
La disquette devrait booter tout seul, charger le fichier en mémoire et retourner au code de celui-ci. 

Vous devriez donc vous retrouver au menu préalablement sauvé de sélection des symboles de code. 
 

 
  



 
 
 

 
 

Si le jeu crash juste après sa mise en mémoire, il y a de forte chance que vous n’ayez pas, comme demandé, attendu un peu  
afin que l’état mémoire se stabilise avant d’effectuer la commande SA 

 

Vous devez donc recommencer l’étape de sauvegarde (commande SA) 

 

- Boot disque de backup          

- Attendre 5secondes après affichage du menu des symboles des protections.    

- Entrer dans l’AR           

- Insérez le disque préalablement crée ‘save1’, taper la commande suivante : SA FULL   

 
 
En attendant 5s avant d’entrer dans l’AR et de taper la commande SA , sur 5 tests je n’ai eu, aucune erreur/crach lors de leur relecture. 

  



Part 10 Test de l’ordre de séquence du code de protection 
 

Normalement, vous devriez donc être sur le menu de protection. 

Dans mon cas, ceci : 
 

 

 

On va déjà vérifier que la routine de vérification fonctionne. Entrer n’importe quel code puis Appuyer sur l’icône OK 

 

 
 

 
 

Bon, tout semble fonctionner normalement, rebooter votre Amiga 
Une fois le menu de protection visible sur l’écran, Entrer dans l’AR et Taper : R 

 

 
  



On va aller jeter un coup d’œil à l’adresse A1, Taper : 
M 7C650 

 
 

Comme prévu, on voie ici la suite de 00 souligné en rouge et en vert, le code ‘normalement’ attendu 00 04 00 10 00 09 

 
Taper : M 7C650 

et changer les 00 soulignés en rouge dans l’image ci-dessus par 00 04 00 10 00 09 

 
 

Taper : X 

Et déplacer maintenant le curseur sur l’icône OK ,puis appuyer sur FIRE. 
 

La protection est alors vérifiée et, comme attendu, on arrive sur le second écran de protection. 

On vient bien donc de passer la 1er phase de protection 😊 

 

 
 

Rebooter votre Amiga. 
Une fois le menu de protection visible sur l’écran, Entrer de nouveau dans l’AR et Taper : M 7C650 

Changer de nouveau les 00 00 mais on va changer l’ordre du code. 
On fonctionne en WORD donc les bons codes sont :  

- 00 04 
- 00 10 

- 00 09 
 

Nous avons déjà essayé dans l’ordre, à savoir 00 04 00 10 00 09. 

Essayer maintenant dans le désordre MAIS TOUJOURS en fonctionnant en ‘WORD’. 
Essayé par exemple : 00 09 00 10 00 04 ou 00 10 00 04 00 09 ou 00 09 00 04 00 10 etc 

Vous pouvez vous amuser à tester toutes les combinaisons si vous le souhaiter. 

Vous pourrez vous apercevoir que dans tous les cas, cela fonctionne. 

La protection semble attendre 3 words (3 symboles), mais peu importe l’ordre dans lequel ils sont entrés. 
Ce qui semble assez logique vue la ‘tête’ de la roue codeuse. (Difficile pour certaine combinaison de savoir quel est le 1er symbole) 

 

  



Part 11 Et la seconde protection alors ? 
 

Afin de travailler plus sereinement sur la seconde protection, nous allons sauver l’état mémoire sur notre disquette save1. 
 

- Laisser donc notre disquette préalablement crée dans le lecteur. 
- Rebooter votre Amiga, laisser le chargement s’effectuer puis, une fois le menu de protection affiché à l’écran, entrer dans l’AR 

- Modifier l’adresse mémoire concernée pour entrer les 3 words comme dans la PART10 
- Retourner maintenant au code du jeu avec l’instruction X 

- Déplacer maintenant le curseur sur l’icône OK 
- Appuyer sur FIRE, le second écran de protection apparait 

- /!\ Attender 5secondes et entrer dans l’AR 
 
Taper :SA FULL 

 

Effectuer un reboot de votre Amiga. 
La disquette devrait booter tout seul, charger le fichier en mémoire et retourner au code de celui-ci comme avant. 

Vous devriez donc vous retrouvez au menu préalablement sauvé de sélection des symboles de code. 
 

On va déjà vérifier que la routine de vérification fonctionne, entrer n’importe quel code puis Appuyer sur l’icône OK 

 

 
 

 
 
Tout semble encore une fois, fonctionner normalement, rebooter votre Amiga. 
Une fois le menu de protection visible sur l’écran, Entrer dans l’AR et Taper : M 7C650 
 

 
 

Comme prévu, on voie ici la suite de 00 souligné en rouge et en vert, le code ‘normalement’ attendu 00 06 00 17 00 16 
  



 

 
Taper : M 7C650 

et changer les 00 soulignés en rouge dans l’image ci-dessus par 00 06 00 17 00 16 

 
 

Taper : X 

Et déplacer maintenant le curseur sur l’icône OK ,puis appuyer sur FIRE. 
 

 
 

Hummm, la copie des 3 words semble ne pas suffire pour la seconde protection. 
 

Rebooter encore votre Amiga. 
Une fois le menu de protection visible à l’écran, Entrer dans l’AR 

 

Normalement, on devrait retrouver notre code au même endroit, dans le doute, taper :  

#FA, alias Find for Addressing Upcode, On va chercher une instruction qui fait appel à une adresse mémoire. 
FA BFE001 

 

 
 

Bon logiquement, ça n’a pas bougé. 
On va donc reposer un breakpoint comme dans la PART04 de ce tuto 

C’est à +8 de la 1er adresse trouvée par la commande FA (02145C+8=21464) 

 
Taper : BS 21464 

Breakpoint inserted Ready. 

 
Taper : X 

A l’aide de notre rouge codeuse, entrer maintenant les bons symboles. 

Dans notre cas en O3 
 

 
 
 

Puis déplacer maintenant le curseur sur l’icône OK , appuyer sur FIRE. 
 

Comme prévu, notre BreakPoint en $21464 étant atteint, on entre automatiquement dans l’AR   



Allons jeter un coup d’œil en mémoire pour voir les changements. 
Taper : M 7C650 

 

 
Changements surlignés en jaune ci-dessus. 

 

 

 
On retrouve comme prévu (voir PART06) le même code copié à deux endroits. 
En  blanc sur fond bleu , le code attendu (déjà présent avant d’utiliser la roue codeuse) 

En  vert sur fond bleu , le code obtenu quand l’on utilise la roue codeuse. 

 
Comme précédemment vue, Data(s) identique  

Le code rentré via roue codeuse = le code déjà présent = Validation du code 
 

 

 
3 symboles entrés donc Data à $03, 3 fois identique. Comme on a ou le voir en PART06. 
 

 

 
La valeur $07 est elle aussi positionné 3 fois, toujours comme à l‘identique en PART06 

Ce qui est logique vue qu’elle correspond toujours au positionnement du curseur (revoir justement PART06 encore 😊) 

  



 
La valeur $01 $2C est elle aussi positionné 3 fois comme dans la PART06 

C’est peut-être la piste que l’on cherche… 
 

 

 
Et le groupe d’inconnue restant. 

 
Néanmoins, la valeur $63 devrait retenir plus notre intention car elle était passé à $69 lors de la 1er phase de protection (PART06) 

$63, $69… deux valeurs assez proches l’une de l’autre finalement. 

 
Les autres valeurs étaient soit présente aussi avant dans la 1er phase de protection, AVANT d’entrer des symboles, 

ou semblent plus aléatoire qu’autre chose… En tout cas, n’ont aucune relation avec ce que l’on a pu déjà voir en PART06. 

 

Donc finalement, ça nous fait plusieurs tests à effectuer 😊 

 
La valeur $01 $2C à recopier 3 fois en (A1+$0E, A1+$4A, A1+$4E) 

La valeur $63 à copier en (A1+$65) 
La valeur $03 en (A1+$05, A1+$45) qui correspond au nombre de symboles entré. 
  



Part 12 Test et validation de la seconde protection 
 

 
Rebooter votre Amiga. 
Une fois le menu de protection visible sur l’écran, Entrer dans l’AR et Taper : M 7C650 et 

- Entrer : 00 17 00 16 00 06 en A1+$3C, A1+$3E et A1+$40 (surligné en jaune dans l’image ci-dessous)  

- Entrer : 63 en A1+$65 (surligné en rouge dans l’image ci-dessous) 
 

/!\ Bien sûr, n’oublier pas de valider en appuyant sur la touche Return chaque ligne modifiée. 
 

 
 
Taper : X 

Déplacer maintenant le curseur sur l’icône OK ,puis appuyer sur FIRE. 

 
La seconde protection est alors vérifiée et… le code continue son chemin. 

On vient bien donc de passer la 2eme phase de protection 😊 

 
Puis sur, comme c’est notre disquette save1 qui est dans le lecteur Amiga et non de backup ou l’original, 

il demande d’insérer la disquette 1 du jeu. 
 

 

 
 
Part 13 Test des deux hacks 
 

Retirer le disque save1 du lecteur, Insérer la disquette de backup Disque1 dans le lecteur. 
Rebooter votre Amiga 

 
Effectuer les ‘hack’ comme précédemment expliqué sur les 2 menus de protection. 
 

Après une super animation d’intro, nous arrivons au niveau 1 
 

Vous pouvez tester le jeu sur plusieurs niveaux si vous le souhaiter, aucun bug ou autre protection en vue. * 

* J’ai testé personnellement jusqu’à la fin du jeu. 
 

Au passage, on notera que le code du niveau qui est affiché quand on perd la partie : EDJI 

 

 
  



Part 14 Création Code pour hack du jeu 

 

On va commencer par créer un code en mémoire avec le jeu chargé. Cela aura pour but de valider qu’il n’y a pas d’autre protection 
comme un check crc.  

On va se servir de notre disquette préalablement crée : save1 
 

 

Rebooter votre Amiga avec la disquette de backup disque 1 dans le lecteur 
Une fois le menu de protection visible sur l’écran, Entrer dans l’AR et taper : 
FA BFE001 

 

 
On retombe sur la même adresse que précédent ce qui est logique mais pas forcément le cas tout le temps, c’est donc mieux de vérifier 😊 
 

D 2145C+8  [RETURN] puis [ESC] 

 

A 2145C+8  JSR 100  [RETURN]  [RETURN] 

 

A 100 

$000100  move.w #1,1f4(a1)  ; On remet le move que l’on a supprimé 

$000106  cmp.b #7,3(a1)  ; On test si le curseur est bien sur OK (position $7) 

$00010C  bne 120   ; Si ce n’est pas le cas, on va en 120 

$00010E  move.w #63,64(a1)  ; Si c’est le cas, on met 63 à l’adresse 64+(A1) 

$000114  move.l 52(a1),3c(a1)  ; on copie le code original de 52+(A1) vers 3C+(A1), on est en longword 

$00011A  move.l 56(a1),40(a1)  ; on copie le reste du code de 56+(A1) vers 40+(A1), on est en longword 

$000120  rts    ; on revient au code original, juste après notre JSR en $100 

$000122     

 

 

 
 

 
On sauve le tout sur notre disquette save1. 

Retirer donc la disquette de backup du disque1 du jeu et insérer la précédente disquette crée save1 
 
Taper : SA FULL puis X 

  



Nous sommes donc revenus au menu de protection. Choisissez 3 symboles au Hazard puis appuyer sur l’icone OK. 
Ok, notre hack fonctionne sur le 1er menu. Faites de même sur le second menu de protection, OK, cela fonctionne aussi. 

Vous pouvez aussi appuyer sur l’icône OK sans sélectionner de symboles. 
 

Comme précédemment, comme c’est notre disquette save1 et non de backup ou l’original, il demande d’insérer la disquette 1 du jeu. 
 

 
Retirer le disque save1 du lecteur, Insérer la disquette de backup Disque1 dans le lecteur et appuyer sur FIRE. 

 
 

Toujours après notre superbe animation d’intro, le jeu se lance. 

 
 

Nous allons vérifier si le code de chargement de niveau fonctionne, Appuyer sur la touche C 
 

Après un court chargement un menu apparait pour entrer les codes de niveau. 

Entrer le code que nous avons déjà noté lors du test du jeu, à savoir EDJI, puis valider le tout en appuyant sur l’icône OK 
 

 
 
Arf…Il semblerait que notre petit patch a altéré le fonctionnement du jeu. 

 
Déplacer le curseur sur l’icône OK puis Entrer dans l’AR et Taper : R 

 

 
 
 

 
 



A1 a toujours la même valeur que précédemment. 
Taper : M 7C650 

 

 
 

Ce qui saute aux yeux en 1er c’est le nombre de champ avec la valeur $FF 
Chose que l’on n’avait pas encore vue précédemment. 

 
Nous allons modifier notre ‘hack’, notre code crée à l’adresse mémoire $100 pour qu’il fonctionne uniquement que sur l’écran 

de sélection des symboles, dans d’autre cas, on remettra le code original en place. 
 

Pour cela, on peut prendre comme référence l’adresse mémoire $7C670 ou $7C680 (surligné en vert dans l’image ci-dessus) 
Auparavant, on avait en $7C670 (donc (A1)+20), une suite de $0000 

 
Retirer le disque de backup Disque1 et Insérer la disquette save1 le lecteur puis rebooter votre amiga. 

Une fois le menu de protection affiché, entrer dans l’AR, nous allons modifier notre code. 

 
Taper : a 100  [RETURN] puis [ESC] 

 

 
 

 

A 100 

$000100  move.w #1,1f4(a1)  ; On remet le move que l’on a supprimé 

$000106  cmp.w  #0000,20(a1)  ; Compare #0000 avec la valeur en (a1)+20  

$00010C  bne  12a   ; pas égale (donc nous sommes sur le menu des levelcode), on va en $12A 

$00010E  cmp.b  #7,3(a1)  ; Sinon on effectue notre hack, voir description plus haut 

$000114  bne  128   ;  

$000116  move.w #63,64(a1)  ;  

$00011C  move.l 52(a1),3c(a1)  ;  

$000122  move.l 56(a1),40(a1)  ;  

$000128  rts     ;  

$00012A  move.l a0,-(a7)  ; On sauve A0 dans la pile 

$00012C  movea.l 4(a7),a0  ; Récupère l’adresse après notre JSR 100 

$000130  suba.l #6,a0   ; On lui enlève 6 qui est la taille de l’upcode ‘JSR 100’ 

$000136  move.l #337C0001,(a0)  ; On a donc l’adresse de notre ‘JSR 100’, on le supprime pour 

$00013C  move.w #1f4,4(a0)  ; remettre le code Original, à savoir : 337C0001F4 

$000142  movea.l (a7)+,a0  ; On remet A0 dans son état initial 

$000146  rts     ; et on revient au code principal 

 

 

 

 



 

Nous allons sauver tout ça sur notre disquette SAVE1 
Taper : SA FULL 

 

On va aussi sauver que le bout de code afin de patcher la disquette de backup. 

Taper : SM patch,100 100+46 

 

Nous allons maintenant vérifier si notre nouvelle modification a fonctionné. 
 

Rebooter votre Amiga, laisser tranquillement le jeu se charger. 
Changer comme vue précédemment de disquette par notre disquette de backup Disque1 

quand le message suivant apparait : 

 

 
 
Normalement, notre hack devrait fonctionner sur les deux écrans de protections avec choix des symboles. 

Une fois dans le jeu, appuyer sur la touche C 
 

Entrer le code que nous avons déjà noté lors du test du jeu, à savoir EDJI, puis valider le tout en appuyant sur l’icône OK 
 

 ! Bingo, tout fonctionne correctement. 

 
Bon il ne reste plus qu’à faire ce hack sur notre disquette de backup ! 

  



Part 15 Hack de notre disquette de backup  

 

Il nous vaut rendre ceci permanent sur notre disquette de Backup Disque1 (qui est toujours dans le lecteur) 
Rebooter votre Amiga puis, entrer sans attendre dans l’AR 

 
Taper : DIR 

 
 
Allons donc voir ce qu’il y a dans le fichier startup-sequence dans le répertoire s 

 
Taper : LM s/startup-sequence, 10000 

 
 
Puis Taper : N 10000 

 
 
C’est donc le fichier another qui est chargé en 1er. Il devrait assez logiquement contenir notre début de code recherché. 

Nous allons le charger en mémoire afin de vérifier ça. 
 
Taper : LM another, 10000 

 
 

On va chercher en mémoire les adresses de code qui font appel à $BFE001, (voir PART04) 

Taper : FA BFE001 

 
 

On vérifie que c’est bien le code déjà vue en PART04 
Taper : D 140AC 

 
 

C’est bien le cas. 
 

  



On va insérer notre JSR 100 
Taper : A 140AC+8 

 

$0140B4  jsr 100  [RETURN]  [RETURN] 

$0140BA 

 

 
 

On sauve le tout. 
Taper : SM another, 10000 17114 

 
 

 

Il nous reste maintenant à mettre notre patch en mémoire $100. 
 

Plusieurs solutions sont possibles : 
- Créer un exécutable et le mettre dans la startup-sequence. 

- Créer une ‘cracktro’ et insérer le patch dedans afin qu’il se loge en adresse mémoire $100 avant l’exécution du fichier another. 

- Faire en sorte que le bootblock fasse tout le travail. 
 

Comme le but de ce tuto n’est pas la création de cracktro, ni d’apprendre l’assembleur, nous allons opter pour la dernière solution : 
Passer par le bootblock 

 
On commence donc par charger celui-ci. 

 
#RT alias Read Track. Permet le chargement de tracks en mémoire 
Taper : RT 0 1 10000  puis  N 10000 

 
 

Afin de mieux apprécier la taille du bootblock, taper : M 10000 

 
  



On en profite pour jeter un coup d’œil au code du bootblock, Taper : D 0 10000+C 

(1000C car c’est ici que commence le code du bootblock, avant ce n’est que ça signature.) 
 

 
 

On va garder ça de côté et le déplacer en mémoire pour plus de facilité. 
#TRANS, alias transfert. Permet le transfert d’une zone mémoire vers une autre. 
Taper : TRANS 10000+c 10000+40 20000 

 
 
Maintenant, nous allons éditer notre bootblock pour ajouter un branchement en 10100 

(10100 car il y beaucoup d’espace libre à cet endroit) 
 
Taper : A 10000+c 

 
 

On va créer un petit bout de code qui va recopier notre patch à l’adresse $100 
Taper : A 10100 

$010100  movem.l d0-d7/a0-a6,-(a7)  ; On sauve tous dans la pile  

$010104  lea  1011E(pc),a0     ; A0 = adresse de notre patch en mémoire 

$010108  lea  00000100,a1     ; A1 = adresse de destination 

$01010E  moveq  #46,d0      ; D0 = taille de data à copier  

$010110  move.b (a0)+,(a1)+     ; + Le début de notre boucle de recopie de notre patch 

$010112  sub.b  #1,d0      ;   On décrémente notre compteur de donnée à copier D0 

$010116  bne  00010110     ; + On boucle 

$010118  movem.l (a7)+,d0-d7/a0-a6  ; On restore le contenu de la pile  

$01012C  rts         ; et on retourne au code original   

$01011E     ...       

 

 
 

 

On insère maintenant en mémoire notre patch préalablement crée et sauvé. 
Insérer la disquette save1 le lecteur puis Taper : LM patch, 1011E 

 
 
 

Sans oublier de replacer nos données préalablement déplacé en $20000 
Taper : TRANS 20000 20040 10010 

 
 

  



Et pour finir, nous allons écrire notre nouveau bootblock sur notre disquette de backup Disque 1 
Insérer notre disquette de backup Disque1 dans le lecteur 

 
#bootchk Alias Boot Check. Permet de calculer un nouveau checksum pour un bootblock 

#WT, alias Write Track. Permet d'écrire une zone mémoire sur la disquette à l'adresse indiqué en cylindre. 
Taper : BOOTCHK  10000   puis  WT 0 1 10000   

 

 
 

 

Rebooter et apprécier le jeu 😊 

 
 

 
 
 

 
 

 
 

 
  



Part 16 Plus loin, plus haut, plus fort  

 

Il est toujours possible d’améliorer un tuto et c’est dans cette optique que cette nouvelle partie voie le jour. 
Si vous avez suivi le tuto, vous savez que :  

- Nous avons modifié le bootblock pour y intégrer notre recopie de hack en $100 
- Modifié le fichier original another pour y intégrer notre jump $100 

 

Le jeu est stocké sur une disquette standard au format AmigaDos et boot ‘normalement’ avec une startup-sequence. 
Dans ce ‘mode de fonctionnement’, on évitera de coder en mémoire basse comme on l’a fait (adresse $100 du patch). 

Ici cela fonctionne sans problème mais ce n’est pas non plus conseillé, bref, une habitude à ne pas prendre. 
De plus, notre hack et routine de recopie de celle-ci se trouve sur le BootSecteur. 

Dans un hack ‘simple’ ça ne pose aucun problème mais si on veut ajouter une petite cracktro, ça va tout simplement virer notre hack 
Il faudra l’intégrer dans la cracktro. 

 
Et puis bon, tout simplement, une nouvelle méthode pour intégrer notre crack est toujours bonne à prendre. 

 
Nous allons créer un petit bout de code sous AsmOne et créer un fichier qui sera exécuté dans la startup-sequence qui va charger 

le fichier original et le patcher. 
 

Cela nous permettra de : 
- D’avoir le BootSecteur de libre (au cas où) 

- De ne pas le fichier original another (plus propre) 

- De ne pas positionner notre crack en $100 (moins dangereux pour le système AmigaDos) 
 

On ne va pas toucher à notre disquette backup déjà cracké. 
Créer une nouvelle disquette de backup avec X-copy, bien sûr en source le disk1 du jeu original du jeu AnotherWorld. 

 

 
 

Charger donc l’assembleur Asm-One, réserver un peu de mémoire chip, 100ko suffira largement. 
 

 
 
Passer en mode édition <ESC>, taper ces quelques lignes de code :  

  



; LoadSeg()  

 

 section seg,code 

 

start: movem.l d1-a6,-(sp) 

 

 move.l 4.w,a6 

 moveq #0,d0 

 lea doslib(pc),a1 

 jsr -552(a6)  ; OpenLibrary() 

 move.l d0,a6 

 tst.l d0 

 beq.s .quit 

 

 lea file(pc),a0  ; name of the file to load and execute 

 move.l a0,d1 

 jsr -150(a6)  ; LoadSeg(file)(d1)=segment d0 

 move.l d0,d7   ; save segment address 

 beq.s .error   ; if segment=0 error 

 

 lsl.l #2,d0   ; # segment *4 

 move.l d0,a0   ; segment address 

 addq.l #4,a0   ; skip hunk size -> start of program 

 

; patch here 

 movea.l a0,a1   ; save A0 

 

 ; put patch here based on A1 

 add.l #$4094,a1  ; (offset) position du patch dans l'exe binaire 

 

 move.w #$4eb9,(a1)+  ; JSR 

 pea Patch(pc)  ; old $100 

 move.l (sp)+,(a1) 

 

; execute here 

 movem.l d7/a6,-(sp)  ; save segment and doslib pointers 

 

 jsr (a0)   ; execute "another" 

  

 movem.l (sp)+,d7/a6  ; get back segment & doslib pointers 

  

 

 move.l d7,d1 

 jsr -156(a6)  ; UnloadSeg(segment)(d1) - stop loaded program 

 

.error move.l a6,a1 

 move.l 4.w,a6 

 jsr -414(a6)  ; close doslib 

 

.quit movem.l (sp)+,d1-a6 

 moveq #0,d0 

 rts 

 

;-------------------------- PATCH Another World (old  $100) 

Patch: 

 move.w #1,$1f4(a1)  ; on remet le move que l'on a supprimé 

 cmp.w #0,$20(a1)  ; 0 en a1+20 ? 

 bne.s .Levelcode  ; non, donc nous sommes sur le menu des levelcode  

 cmp.b #7,3(a1)  ; sinon on effectue notre hack 

 bne.s .exit 

  

 ; hack 

 move.w #63,$64(a1) 

 move.l $52(a1),$3c(a1) 

 move.l $56(a1),$40(a1) 

.exit rts 

 

 ; menu levelcode 

.Levelcode 

 move.l a0,-(a7)  ; sauve a0 dans la pile 

  

 movea.l 4(a7),a0  ; recupere l adresse après notre JSR $100 

 suba.l #6,a0   ; on enleve 6 (taille du jsr $100) 

 move.l #$337c0001,(a0)  ; remet le code original 

 move.w #$1f4,4(a0)  ; 

 

 movea.l (a7)+,a0  ; remet a0 dans son etat initial 

 rts 

 

doslib dc.b "dos.library",0 

 even 

file dc.b "another",0 

 even 

 



Bien sûr, vous pouvez sauver le code source sur votre disquette save1 (par exemple). 

Insérez notre nouvelle disquette de backup fraichement copié dans le lecteur. 
 
 

Basculer en mode ligne de commande avec la touche <ESC>, Taper a pour assembler puis, WO DF0:anotherstarter 

pour sauver le tout. 

 

 
 
 

Il nous reste maintenant à modifier le fichier s/startupt-sequence pour exécuter notre nouveau fichier/patch anotherstarter 

Entrer dans l’AR, taper :  
 
n 10000 puis  anotherstarter 

 

On peut vérifier avec la commande, m 10000 

Et on sauve le tout, taper :  sm 0:/startup-sequence, 10000 1000E 

 

 
 

Rebooter et encore une fois, apprécier le jeu 😊 

 

 
PS : Un grand merci à Heavy aka VrS!-Cybfree pour l’idée et le code de cette dernier ‘Part’. 


